
ABSTRACT 
This paper lists and demonstrates the ODS techniques that I have found most useful in creating user friendly PDF 
reports in SAS® 9.2 on a Windows operating system. Basic ODS techniques include creating PDF files (e.g. ODS 
PDF FILE= STYLE= KEYWORD= COLUMNS=), modifying page setup (e.g. orientation, date, title, and footnote), 
and saving procedural output to data sets (e.g. ODS OUTPUT), The examples also demonstrate simultaneously 
creating two PDF destinations (i.e. ODS PDF ID=) and selectively including or excluding procedural output to each 
destination (e.g. ODS PDF SELECT, EXCLUDE and WHERE=).  Two PDF files are simultaneously created where 
PROC FREQ listings are directed to one file and selected PROC FREQ results are combined with PROC REPORT 
to create an executive summary in the other PDF file.   

INTRODUCTION 
The goal of this paper is to demonstrate the techniques I have used most frequently to produce nice looking PDF 
files containing output from SAS procedures. This paper is not an introduction to ODS, ODS PDF, or PROC 
REPORT. Featured capabilities are simultaneously creating more than one PDF file and including only selected 
parts of PROC FREQ results in one or the other files. For example, screen shots of two PDF files that were created 
simultaneously are shown in Figures 1 and 2. Several FREQ procedures were submitted that generated relatively 
standard output to the PDF shown in Figure 1 (only the first output is shown) while Figure 2 shows selected output 
displayed with PROC REPORT.  

Figure 1 PDF file with PROC FREQ output Figure 2 PDF file with PROC FREQ restuls Displayed 
with PROC REPORT 

 

 

 

 

The paper begins by demonstrating the most basic ODS PDF syntax and builds on that to finally demonstrate all the 
syntax necessary for producing the PDF files shown in Figures 1 and 2. The sections that follow discuss: (a) creating 
a PDF file, (b) styling titles and footnotes, (c) capturing results from procedures as data, (d) sending specific parts of 
a procedure’s results to a destination, (e) simultaneously creating multiple PDF files, and (f) displaying the results 
with PROC REPORT. 

CREATING A PDF FILE 
The examples in this paper use the services data set created with the syntax shown in Figure 3.  Figure 4 shows the 
basic syntax needed to create a PDF. The FILE= option is used to define its full path and name. Every PDF file in 
this paper is created with an initial ODS PDF FILE= line, followed by procedural syntax and a final ODS PDF CLOSE

line.  I will use the macro variable SASEXPORT to represent the full path to the location where the PDF file is 
created, for example, “c:\ .”  The table created in the PDF file by ODS PDF and PROC FREQ is shown in Figure 4. 

Figure 3 Serives Data set 
PROC FORMAT; 
  VALUE yn 1 = 'Yes' 0 = 'No'; 
RUN; 
DATA services ; 
LABEL  
t1_1 = 'Caregiver Mental Health Need' 
t1_2 = 'Caregiver Alcohol and Drug Service 
Need' 
t1_3 = 'Medical Insurance Coverage' 
t1_4 = 'Access to Routine Medical Care' 
t1_5 = 'Last Medical Checkup' 
t1_6 = 'Last Dental Checkup' 
t2_1 = 'Caregiver Mental Health Need' 
t2_2 = 'Caregiver Alcohol and Drug Service 
Need' 
t2_3 = 'Medical Insurance Coverage' 
t2_4 = 'Access to Routine Medical Care' 
t2_5 = 'Last Medical Check' 
t2_6 = 'Last Dental Check'; 
FORMAT t1: t2: yn.; 
ARRAY a t1_1- t1_6 t2_1- t2_6 ; 
DO k = 1 TO 100; 
DO OVER a; 
   a = ROUND(RANUNI(9033905)) ; 
END; 
OUTPUT; 
END; 
RUN; 

 

Figure 4 Creating a PDFfile using FILE= 
 

%LET sasexport = c:\; 
ODS PDF 
FILE="&sasexport.example1.pdf"; 
PROC FREQ DATA=services; 
TABLE t1_1; 
RUN; 
ODS PDF CLOSE; 

 

 

The following table shows some of the ODS PDF options that will be discussed in the in this section on creating a 
PDF file. 

Table 1 Capability of Essential Options used on the ODS PDF Statement 
 

Capability 

 

Options 

 
Alter the overall style of the PDF file, e.g. background color 

STYLE =  

Add metadata to the PDF file that may be viewed in the folder of a Windows OS 
 
AUTHOR=, TITLE=, 
SUBJECT=, KEYWORDS= 

 
Prevent procedures from creating a new page, if space allows. e.g. 
STARTPAGE=NEVER 

STARTPAGE= 

 
Allow procedural output to be placed in more than on column on a page  
 

COLUMNS= 

Allow more than one PDF file to be created at once 
 

ID= 

 
CONTENTS=yes creates a table of contents at the beginning of the PDF file 
 

CONTENTS= 

ODS PDF STYLE= 
Perhaps there is no greater evidence of the power of ODS PDF options than the use of STYLE= to change the entire 
look of the PDF file and the procedural output therein. Figure 5 shows the use of PROC TEMPLATE to list the styles 
that are available to use in the STYLE= option.  The syntax produced a list of styles that appeared in the output 
window and I chose to use the OCEAN style. As shown in Figure 5, I added STYLE=OCEAN to the ODS PDF 
statement, and the background color, font type and size were changed from the default style.  

The ability to customized a PDF using STYLE= may also be greatly expanded by creating your own style based on 
existing styles.  In Figure 6, I create the style MYOCEAN from the OCEAN style, and replace some of the table style 
attributes in OCEAN with my own.  Using PROC TEMPLATE to define styles and change attributes takes practice, 
but a little syntax goes a long way in making changes (see Gupta’s and Haworth’s paper in Recommended 
Reading). 

Figure 5 Using STYLE= to Change the Over Look 
 
PROC TEMPLATE; 
 LIST STYLES; 
RUN; 
 
ODS PDF FILE="&sasexport.example2.pdf" 
STYLE=OCEAN; 
PROC FREQ DATA=services; 
TABLE t1_1; 
RUN; 
ODS PDF CLOSE; 

 

 

 

 

 

Figure 6 Creating your own style 
 

PROC TEMPLATE;  
DEFINE STYLE myocean; PARENT= 
styles.ocean; 
REPLACE TABLE FROM OUTPUT / 
FRAME = void  
RULES = rows  
CELLPADDING = 3pt  
CELLSPACING = 0.0pt  
BORDERWIDTH = 0.2pt; 
END; 
RUN;  

 

ODS PDF FILE="&sasexport.example2a.pdf" 
STYLE=myocean; 
PROC FREQ DATA=services; 
TABLE t1_1; 
RUN; 
ODS PDF CLOSE; 

 

 

 

ODS PDF KEYWORDS= 
Several options are available to add metadata to your PDF.  One value of adding metadata is the ability to have a 
nice descriptive name for your files while also being able to organize your files on other dimensions, such as the 
name of the program that created your file, the person that requested the file, the reason for the file (e.g. quarterly 
report), or the phase of the project when the file was generated (e.g. follow-up).    

Figure 7 shows the KEWORDS column in my Windows folder. In this example, I’ve given a descriptive name to the 
PDF file “Ocean Style PDF with KEYWORDS” and you can see that the keywords column shows the name of the 
program that created the file and the descriptor “example3.”  The syntax in Figure 7 was used to add the metadata 
using the KEYWORDS option in ODS PDF line.  I obtained the name of the program from an environmental variable 
SAS_EXECFILENAME which is available in the enhanced editor. %SYSGET assigned the name of the program to 
the macro variable PROGRAMNAME. 

Figure 8 shows the last step you need to take just once to use the metadata. I changed the default columns in my 
Windows folder to add the keywords column.  I selected VIEW, CHOOSE DETAILS from the folder menu to obtain 
the dialog in Figure 8 and then clicked on keywords. 

Figure 7 Windows Folder with Keywords as a Column 

 
 

 

%LET programname =  %SYSGET(SAS_EXECFILENAME); 
%LET mytitle = Ocean Style PDF with KEYWORDS; 
ODS PDF FILE="&sasexport.&mytitle..pdf"  
STYLE=OCEAN KEYWORDS="&programname example3"; 
PROC FREQ DATA=services; 
TABLE t1_1; 
RUN; 
ODS PDF CLOSE; 

 

Figure 8  Adding Keywords Column to 
Windows Folder 
 

 

 

CONSOLIDATION OF A PDF FILE WITH VARIOUS OPTIONS 
 By default PROC FREQ does not maximize the use of space in the PDF output file. Figure 9 shows PROC FREQ 
with the AGREE option and pseudo code representing the submission of three additional FREQ procedures. Each 
PROC FREQ took one page for the cross-tabulation results (e.g. Figure 10) and one page for the McNemar’s and 
Kappa results (e.g. Figure 11).  I ran four PROC FREQs and produced an 8 page PDF, the first page is shown in 
Figure 10 and eighth page in Figure 11. 

Figure 9 Four PROC FREQ Results to 
PDF  
 

 
ODS PDF 
FILE="&sasexport.example4.pdf" 
STYLE=OCEAN; 
PROC FREQ DATA=services; 
TABLE t1_1 * t2_1/AGREE; 
RUN; 
[repeat PROC FREQ for 3 more 
variable pairs] 
ODS PDF CLOSE; 

 

Figure 10 Page One of Eight 
Pages of PROC FREQ 
 

 

Figure 11 Page Eight of PROC 
FREQ Output 

With just a few options I was able to produce the same output in just 2 pages rather than 8. Figure 12 shows two 
PROC FREQs and the resulting output on a single page. The COLUMNS=2 option was used to allow the PROC 
FREQ results to be written side-by-side; however, the output was too long for the page, so I used the general ODS 
option NOPROCTITLE to remove the default title “The Frequency Procedure.”  The PROC FREQ option NOTITLE 
was used to remove the title above the McNemar’s table, “Statistics for Table of…”  

Figure 12 Using Options COLUMNS=, NOPROCTITLE, and NOTITLE 
ODS PDF FILE="&sasexport.example4b.pdf" STYLE=OCEAN COLUMNS=2; 
ODS NOPROCTITLE; 
PROC FREQ DATA=services NOTITLE; 
TABLE t1_1 * t2_1/AGREE; 
RUN; 
PROC FREQ DATA=services NOTITLE; 
TABLE t1_2 * t2_2/AGREE; 
RU 
ODS PDF CLOSE; 

 

 

 

By default, PROC FREQ creates a new page in ODS PDF, and in many cases ODS PDF STARTPAGE= NEVER is 
very useful in allowing more than on procedure output to appear on a page. It was not needed in this example 
because when COLUMNS=2 is set the page break created by the procedure causes the next procedure output to go 
into the next column. If there had been room for more than one procedure output per column, then ODS PDF 
STARTPAGE=NEVER would have been useful. 

STYLING TITLES AND FOOTNOTES 
Table 2 shows several options and style attributes that I have found essential for placing and styling the titles and 
footnotes in PDF files.  

Table 2 Capability of Essential OTIONS and Style Attributes 
 
Capability Options/Style Attributes 

 

System options may be used to remove the default SAS date and page 
number, and to make the page PORTRAIT or LANDSCAPE. 

 

 

NODATE, NONUMBER, 
ORIENTATION=  

An option of ODS that is instrumental in obtaining the page number from 
Acrobat.  Defines a character that tells ODS to treat the text to follow the 
character as an Acrobat command. 

 

ODS ESCAPECHAR= 

 

 

Style attributes that may be used in TITLE and FOOTNOTE statements HEIGHT=, J=, F= 

 

 

Figure 13 shows syntax using the options and style attributes along with a small macro program I used to generate a 
formatted time and date. Figure 14 shows the title and footnotes created in the PDF file. 

Figure 13 Using Options and Style Atributes to modify titles and footnotes 
OPTIONS ORIENTATION=LANDSCAPE NODATE NONUMBER; 
TITLE1 HEIGHT=12PT J=LEFT F=ARIAL "Modifying OPTIONS"; 
%macro repnow ; 
  %local d t ; 
  %let d = %sysfunc( date( ), weekdate29 ); 
  %let t = %sysfunc( time( ), timeampm8 ); 
  &t  &d 
%mend  repnow; 
%let ff = HEIGHT=8PT J=RIGHT F=ARIAL; 
FOOTNOTE1 &ff "Page ^{thispage} of ^{lastpage}"; 
FOOTNOTE2 &ff "%repnow"; 
 
ODS PDF FILE="&sasexport.example5.pdf” STYLE=OCEAN; 
ODS ESCAPECHAR='^‘; ODS NOPROCTITLE; 
PROC FREQ DATA=services NOTITLE; 
TABLE t1_1 * t2_1/AGREE; 
RUN; 
ODS PDF CLOSE; 
 

Figure 14 Example PDF with Custom TITLE and FOOTNOTE 

CAPTURING RESULTS FROM PROCEDURES AS DATA 
It is necessary to know the names of the objects that procedures make available to ODS in order to use the object to 
save the results of the procedure to a data set. Knowing the names of the objects also allows the capability of saving 
specific parts of a procedure’s output to a PDF while excluding others (e.g. ODS PDF SELECT WHERE). We will 
cover this in the next section. Here we will determine the names of the PROC FREQ objects in our example that are 
available to ODS and save those results to data sets.  

The names of the objects may be obtained with the use of ODS TRACE.  In this example, prior to creating a PDF 
file, I submitted the following: 

ODS TRACE ON; 
PROC FREQ DATA=services; 
TABLE t1_1 * t2_1; 
RUN; 
ODS TRACE OFF; 

 

In addition to creating PROC FREQ output in the list window, the following was written to the log: 

Name:       CrossTabFreqs 
Label:      Cross-Tabular Freq Table 
Template:   Base.Freq.CrossTabFreqs 
Path:       Freq.Table1.CrossTabFreqs 
 

The name of the object available to ODS is CROSSTABFREQS.  It is important to note that not only do other 
procedures make available different objects, but depending on a procedure’s statements and options objects may or 
may not be available.  As a result, it is important to use ODS TRACE with the exact procedural syntax that you intend 
to use.  For example, when I used ODS TRACE with PROC FREQ and the AGREE option on the TABLE statement, I 
obtained the names of two additional objects MCNEMARSTEST and SIMPLEKAPPA.  The three objects of 
CROSSTABFREQS, MCNEMARSTEST, and SIMPLEKAPPA correspond to the 3 tables shown earlier in Figure 12. 

I submitted the following syntax in order to save the results of two of the objects, CROSSTABFREQS and 
MCNEMARSTEST, to data sets. The object names are shown on the left of the equal signs in the ODS OUTPUT 
statement. I chose to name the data sets, on the right of the equal signs, with the same names: 

ODS OUTPUT CROSSTABFREQS= CrossTabFreqs MCNEMARSTEST = McNemarsTest; 
PROC FREQ DATA=services; 
TABLE t1_1 * t2_1/AGREE; 
RUN; 
ODS OUTPUT CLOSE; 
 

This syntax does not create a PDF; however, saving the results of an object to a data set may be done in order to 
later display the results in a PDF. Or as in later examples, it may be done during the actual creation of a PDF in 
order to immediately present the results with another procedure, such as PROC REPORT. In either case, the 
selected inclusion or exclusion of results is powerful. The next section shows the use of ODS PDF SELECT in order 
to further extent this capability.     

SENDING SPECIFIC PARTS OF PROCEDURE OUTPUT TO A PDF FILE 
In addition to ODS OUTPUT as a means of capturing and selectively presenting results from procedures, the ODS 
PDF SELECT/EXCLUDE and WHERE options allow selected procedural output to be sent to or excluded from a 
PDF file.  The following syntax uses the name of the CROSSTABFREQS object in an ODS SELECT statement to 
cause only the first table produced by the PROC FREQ to be sent to the PDF file: 

ODS PDF FILE="&sasexport.Example 6.pdf" …; 
ODS PDF SELECT WHERE=(_NAME_ in ('CrossTabFreqs')); 
PROC FREQ DATA=services; 
TABLE t1_1 * t2_1/AGREE; 
RUN; 
ODS PDF CLOSE; 
 

All three tables, cross tabulation, McNemar’s and Simple Kappa results were sent to the list output; however, only 
the cross tabulation was sent to the PDF file due to the selection of the CROSSTABFREQS object in the ODS PDF 
SELECT statement. 

SIMULTANEOUSLY CREATING TWO PDF FILES AND SELECTIVELY SENDING RESULTS 
The use of the ID= option further extends the capability of selecting and excluding results by allowing 2 or more PDF 
files to be created simultaneously.  In the following example, the PROC FREQ results are used in both PDF files, but 
the file identified as ID=LONG contains all the results from the PROC FREQ while the PDF identified as ID=SHORT 
does not contain the Simple Kappa table.    

ODS PDF (ID=long) FILE="&sasexport.Example 7a.pdf"; 
ODS PDF (ID=short) FILE="&sasexport.Example 7b.pdf" STYLE=OCEAN; 
ODS PDF (ID=short) EXCLUDE WHERE=(_name_ in ('SimpleKappa'));  
PROC FREQ DATA=services NOTITLE; 
TABLE t1_1 * t2_1/AGREE; 
RUN; 
ODS PDF (ID=short) CLOSE; 
ODS PDF (ID=long) CLOSE;  
 

The names of ID= are of your choosing.  All the ODS PDF options are applied separately for each PDF file, so for 
example, I chose to format the SHORT PDF with STYLE=OCEAN, while leaving the LONG PDF with default style.  It 
is necessary to close both PDF destinations. 

It may also be useful to create two PDF files at once, selectively including or excluding results, while also saving the 
results using ODS OUTPUT in order to display those results with a nicer way using PROC REPORT. The following 
syntax combines all three techniques of ID=, SELECT/EXCLUDE WHERE, and OUTPUT. I have inserted pseudo 
code “[Insert …]” to indicate that syntax including PROC REPORT would be used to present the results of the PROC 
FREQ. The results found in the data sets CROSSTABFREQS and MCNEMARSTEST: 

ODS PDF (ID=lis) FILE="&sasexport.Example 8a.pdf" ; 
ODS PDF (ID=nice) FILE="&sasexport.Example 8b.pdf" STYLE=OCEAN; 
ODS PDF (ID=nice) EXCLUDE _ALL_;  
ODS OUTPUT CROSSTABFREQS= CrossTabFreqs (rename=(t1_1=var1 t2_1=var2))  
MCNEMARSTEST = McNemarsTest; 
PROC FREQ DATA=services; 
TABLE t1_1 * t2_1/AGREE; 
RUN; 
ODS OUTPUT CLOSE; 
ODS PDF (ID=lis) EXCLUDE _ALL_;  
ODS PDF (ID=nice) SELECT WHERE=(_NAME_ in (‘Report’)); 
[Insert DATA step, PROC SQL, and PROC REPORT syntax to follow] 
ODS PDF (ID=lis) CLOSE;  
ODS PDF (ID=nice) CLOSE; 

     

The LIS and NICE PDF files are started on the first two lines. NICE is formatted with OCEAN style. The third line 
immediately excludes all output from NICE. This will stay in effect until it is changed.  As a result of the EXCLUDE, 
the PROC FREQ output will not go to NICE, but it will go to the LIS PDF.  The ODS OUTPUT statement is used to 
create the CROSSTABFREQS and MCNEMARSTEST data sets that store the results from the PROC FREQ.  
Following the PROC FREQ and the ODS OUTPUT CLOSE statement, all output is excluded from LIS using the 
statement: ODS PDF (ID=lis) EXCLUDE _ALL_;. LIS has its output already. NICE PDF will receive its output via 
PROC REPORT using the results in CROSSTABFREQS and MCNEMARSTEST.  

In order to maximally consolidate the results, I elected to report the McNemar’s results as a line under my PROC 
REPORT table. Figure 15 shows that the MCNEMARS data set has 3 observations.  I wanted to list the value of 
LABEL1 and CVALUE1 as for example:  Statistic (S)=5.6667 DF=1 Pr > S=0.0173. 

Figure 15 The MCNEMARS data set 

In order to capture the McNemar’s test results I used following PROC SQL syntax to save the values of LABEL1 and 
CVALUE1 to a macro variable called MC.  The CATX function placed an “=” sign between the values of the variables 
and the key phrase SEPARATED BY “ “ caused the values of the variables from all three observations of 
MCNEMARS to be saved to the MC macro variable. The line %PUT &MC resolved the macro variable to the log 
where I reviewed its contents as: Statistic (S)=5.6667 DF=1 Pr > S=0.0173 . 

PROC SQL NOPRINT;  
SELECT CATX('=',label1,cvalue1) into : mc SEPARATED BY ' ' FROM McNemarsTest; 
QUIT; 
%PUT &mc; 

 

Since the McNemar’s test compares the marginal 
percent in Figure 16 at time 1 (e.g. 36% Yes) to that at 
time 2 (e.g.53% Yes), I used a DATA step to extract 
only the marginal results from CROSSTABFREQS. 

DATA CrossTabFreqs; 
SET CrossTabFreqs; 
IF var1 = . THEN onevalue = var2; 
ELSE IF var2 = . THEN onevalue = var1;  
IF (_type_ ='01' AND var1 =.) OR 
(_type_ ='10' AND var2 =. ); 
RUN;  

 

Figure 17 shows the CROSSTABFREQS after the 
DATA step. 

Figure 17 PROC PRINT of CROSSTABFREQS with only Marginal N and Percent 

 

In addition to keeping only the marginal results (Figure 17), I also created the variable ONVALUE. Since the 
marginal results for each variable are on different rows, my plan was to use PROC REPORT to create just 2 rows of 
results rather than four.  The following PROC REPORT syntax shows that I did this by defining ONEVALUE as a 
GROUP variable and _TYPE_ as an ACROSS variable.  This strategy created 2 rows, one Yes and one NO, and a 
frequency and a percent column for _TYPE_ =10 and for _TYPE_ =01.  The value 10 for type identified the time 1 
results and 01 identified the time 2 results, so I created the format $TIMETYPE. I also applied the YN format used for 
the original variables to ONEVALUE. The macro variable MC was resolved in the LINE statement within the AFTER 
_PAGE_ compute block in order to show the results of the McNemar’s test at the bottom of the table. 

PROC FORMAT; 
   VALUE $timetype '10'= 'Time 1' '01'='Time 2';  
RUN; 
PROC REPORT DATA=CrossTabFreqs nowd MISSING; 
   COL ("Table 1 Caregiver Mental Health Need" onevalue (_TYPE_), (frequency 
 percent)); 
  DEFINE onevalue /GROUP 'Responses' STYLE=[CELLWIDTH=200]  CENTER; 
  DEFINE _TYPE_ /ACROSS FORMAT=$timetype. ""; 
  DEFINE FREQUENCY/ 'n' STYLE=[CELLWIDTH=70]; 
  DEFINE PERCENT/'%' STYLE=[CELLWIDTH=70] FORMAT=5.1; 
  RBREAK AFTER / OL SKIP SUMMARIZE SUPPRESS; 
  COMPUTE AFTER _PAGE_ /; 
    LINE “&mc"; 
  ENDCOMP; 
  FORMAT onevalue yn.; 
  RUN; 
 

I used a couple techniques that are particularly nice when placing many PROC REPORT output on a single page. 
First, it is very useful to assign the table title in the COL statement because the title will always stay with the 
REPORT table. STYLE=[CELLWIDTH=] is also very useful because it can be used to make all the columns of all the 
tables on a page a constant width. 

Figure 16 Identifying the Marginal n and Percent 

CONCLUSION 
This paper reviewed basic ODS techniques for creating PDF files (e.g. ODS PDF FILE= STYLE= KEYWORD= 
COLUMNS=), modifying page setup (e.g. orientation, date, title, and footnote), and saving procedural output to a 
data set (e.g. ODS OUTPUT). It also demonstrated simultaneously creating two PDF destinations (i.e. ODS PDF 
ID=) and selectively including or excluding procedural output to each destination (e.g. ODS PDF SELECT, 
EXCLUDE and WHERE=).    

